


FICHE 2 EXCEL : Références relatives/absolues et nommer les cellules

Notez-le bien

Par défaut l'adressage des cellules dans les formules est relatif. Par exemple, si je veux calculer le montant pour plusieurs articles, à partir des prix unitaires et des quantités, je ne vais écrire qu'une fois la formule, puis la recopier.

 Testez ce petit exemple (saisir **une seule fois la formule** et la recopier) et vérifiez que les formules correspondent bien au tableau du dessous.


	Quantité	Prix unitaire	Montant
Clé USB Corsair B2.0-16Go	10	13,95 €	139,50 €
Clé USB Corsair Flash-128Go	5	64,95 €	324,75 €
Carte mémoire Kingston 32Go	1	27,95 €	27,95 €
Boite 10 CD-R 700Mo Verbatim	2	14,95 €	29,90 €

	A	B	C	D
1		Quantité	Prix unitaire	Montant
2	Clé USB Corsair B2.0-16Go	10	13,95	=B2*C2
3	Clé USB Corsair Flash-128Go	5	64,95	=B3*C3
4	Carte mémoire Kingston 32Go	1	27,95	=B4*C4
5	Boite 10 CD-R 700Mo Verbatim	2	14,95	=B5*C5

Le montant dépend bien pour chaque ligne de la quantité et du prix unitaire ; c'est en tout cas ce qu'en a déduit Excel, lorsque vous avez effectué la recopie (et il avait raison ;-). Par défaut, lors d'une recopie, les références aux cellules se changent automatiquement. **On parle de références relatives.**

Les références absolues, pour quoi faire ?

Voici un exemple simple : votre boulangère a peut-être un petit carton sur le comptoir, avec le prix de 2, 3, 4... 10 baguettes ; cela lui évite de se creuser la tête chaque fois qu'un client lui demande x baguettes.

 Testez ce petit exemple. Si vous ne saisissez qu'une seule fois la formule et la recopiez, vous vous heurtez forcément à un problème... celui représenté ci-dessous (à gauche). Observez les formules et vérifiez qu'elles sont similaires à celles ci-dessous (à droite).

	A	B
1	Prix de la baguette	0,75 €
2		
3	Quantité	Montant à payer
4	1	0,75 €
5	2	0,00 €
6	3	#VALEUR!
7	4	3,00 €
8	5	0,00 €
9	6	#VALEUR!
10	7	21,00 €
11	8	0,00 €
12	9	#VALEUR!
13	10	210,00 €

	A	B
1	Prix de la baguette	0,75
2		
3	Quantité	Montant à payer
4	1	=A4*B1
5	2	=A5*B2
6	3	=A6*B3
7	4	=A7*B4
8	5	=A8*B5
9	6	=A9*B6
10	7	=A10*B7
11	8	=A11*B8
12	9	=A12*B9
13	10	=A13*B10

En effet, Excel a cru bon de recopier les formules en considérant que la référence aux cellules était relative, mais cette fois, il s'est trompé...

Ce que vous souhaitez, c'est que la référence à la quantité soit relative, mais que le prix de la baguette soit toujours celui de la cellule B1.

☒ D'où provient le #VALEUR! ?

Pour que B1 reste bien B1, et ne devienne pas B2 B3 B4..., il faut fixer la référence à la cellule de manière à ce que cette référence soit **absolue**.

Pour cela, il faut ajouter des \$ devant la référence de la cellule B1.

☒ Procédez de la manière suivante : supprimez le contenu des cellules B4 à B13 ; en B4 re-saisissez la formule comme précédemment, mais au lieu de B1, **saisissez \$B\$1** ; effectuez la recopie et constatez cette fois-ci que le résultat est correct. Vérifiez que si le prix passe à 0,80 €, le re-calcul se fait bien !

	A	B
1	Prix de la baguette	0,75 €
2		
3	Quantité	Montant à payer
4	1	0,75 €
5	2	1,50 €
6	3	2,25 €
7	4	3,00 €
8	5	3,75 €
9	6	4,50 €
10	7	5,25 €
11	8	6,00 €
12	9	6,75 €
13	10	7,50 €

	A	B
1	Prix de la baguette	0,75
2		
3	Quantité	Montant à payer
4	1	=A4*\$B\$1
5	2	=A5*\$B\$1
6	3	=A6*\$B\$1
7	4	=A7*\$B\$1
8	5	=A8*\$B\$1
9	6	=A9*\$B\$1
10	7	=A10*\$B\$1
11	8	=A11*\$B\$1
12	9	=A12*\$B\$1
13	10	=A13*\$B\$1

☺ **Astuce** : Un moyen simple de mettre des \$ consiste à appuyer une fois sur la **touche F4** après avoir saisi la référence de la cellule (ou avoir pointé la cellule avec la souris).

👤 L'adressage mixte, what is it ?

Il est possible, non seulement de fixer complètement la cellule, mais également de ne fixer que la colonne ou que la ligne. Pour ce faire, il ne faudra mettre le \$ que devant la lettre (pour fixer la colonne) ou que devant le chiffre (pour ne fixer que la ligne).

☞ Voici un autre exemple qui vous permettra de comprendre :

	A	B	C	D	E
1	Réductions				
2	Quantité	1 à 10	11 à 20	21 à 50	au-delà de 50
3	Taux remise	0%	10%	16%	20%
4					
5					
6	Prix unitaires nets				
7	Quantité	1 à 10	11 à 20	21 à 50	au-delà de 50
8					
9	Disquettes 3"1/2	0,35 €	0,32 €	0,29 €	0,28 €
10	CD-W	0,95 €	0,86 €	0,80 €	0,76 €
11	CD-RW	1,55 €	1,40 €	1,30 €	1,24 €
12	Ramette 250 feuilles A4	4,50 €	4,05 €	3,78 €	3,60 €
13	Ramette 250 feuilles A3	5,20 €	4,68 €	4,37 €	4,16 €

Les zones en gris sont des zones calculées

En C9, la formule pourrait être =B9*(1-C3)

1^{ère} erreur : si on recopie vers la droite, le prix de base ne restera pas fixe !

2^{ème} erreur : si on recopie vers le bas, le taux de remise ne sera plus valable !

Par conséquent, si on veut optimiser l'élaboration d'une telle feuille de calcul, on cherchera à trouver la bonne formule qui pourra être recopiée sur toute la zone :

- pour le prix de base des articles : il faut fixer la colonne, mais pas la ligne → \$B9
- pour la réduction : il faut fixer la ligne, mais pas la colonne → C\$3

⇒ D'où la formule suivante : **=\$B9*(1-C\$3)**

 Testez cette formule

☺ Astuce : Fixer seulement la ligne ou la colonne peut également se faire grâce à la touche F4. Après avoir saisi la référence de la cellule (ou avoir pointé la cellule avec la souris), appuyez 2 fois sur F4 pour ne fixer que la ligne ou 3 fois pour ne fixer que la colonne. Appuyer 4 fois sur la touche F4 revient à l'adressage relatif : en fait, il s'agit d'un cycle (modulo 4).


Une cellule peut avoir un nom (vous en avez bien un, vous ;-)

Nommer une cellule permet d'utiliser son nom au lieu de sa référence pour la désigner.

Pour nommer une cellule, cliquer sur cette cellule, puis cliquer sur la zone nom qui contient sa référence.

Lui donner un nom (*le trait de soulignement est autorisé, mais pas le trait d'union, ni l'espace*).

Attention ! C'est la cellule qui contient la valeur qu'il faut nommer, pas celle qui contient l'intitulé.



 Testez l'exemple ci-contre :


- Nommer HT la cellule B1
- Nommer TAUX la cellule B2
- Taper la formule en B3 en pointant les cellules B1 et B2
□ en fait la formule va utiliser les noms que vous avez donnés



Remarquez la liste de tous les noms mis en place dans un classeur → liste déroulante à droite de la zone.

□ Pour supprimer un nom

- Il faut passer par le Gestionnaire de noms (onglet Formules)
- Sélectionner le nom à supprimer, puis cliquer sur le bouton




Supprimer

Attention :



Un nom est défini pour l'ensemble d'un classeur, et pas uniquement pour la feuille courante.

□ Un nom se comporte comme un adresse absolue

 Recommencez l'exemple sur les références absolues (effacer les formules) :

- Nommer la cellule qui contient le prix de la baguette (cellule B1)
- Saisir la formule sur la 1^{ère} ligne : quand vous pointez la cellule B1, c'est son nom qui s'inscrit dans la barre de formule.
- Recopier la formule vers le bas : on obtient le même résultat qu'avec les \$.

Attention :



Une cellule que vous avez nommée se comportera toujours en tant que référence absolue lorsque vous la pointerez au cours de l'élaboration d'une formule. Pour conserver le *référencement* relatif, vous pouvez toujours taper sa référence.

APPLICATIONS

Pour savoir si tout a été compris, réalisez les petits exercices suivants, dans différents onglets de votre classeur : Vous nommerez les onglets (par double-clic sur l'onglet) en Fabrication Dossier, Table Multiplication,

☛ Sur les modèles de feuilles de calcul fournies, la légende est la suivante :

- 🕒 Les zones grisées correspondent à des cellules calculées.
- 🕒 Les cellules en gras indiquent les formules initiales, qui sont recopiées ensuite sur toute une zone.

☛ **Objectif : Utiliser au maximum la recopie des formules, vous devez utiliser par conséquent les références absolues quand c'est nécessaire (nommez alors les cellules), voire l'adressage mixte.**

Application n° 1 : Fabrication des dossiers

Remarque : Pour la zone D4:D5 vous fusionnez les 2 cellules D4 et D5

	A	B	C	D	E	F
1	FABRICATION DES SUPPORTS DE COURS					
2						
3		Temps (en secondes)		Nombre Exemplaires		
4	Page	0,5		25		
5	Couverture	6				
6	Intercalaire	3				
7						
8	Dossier	Nombre de pages	Nombre de couvertures	Nombre d'intercalaires	Temps total en secondes pour 1 exemplaire	Temps total en min pour tous les exemplaires
9	Module 1	16	1	0	14	5,83
10	Module 2	30	2	6	45	18,75
11	Module 3	50	2	8	61	25,42

B4, B5, B6 et D4 seront nommées

Les zones grisées sont les zones calculées

Application n° 2 : Table de multiplication

ATTENTION : Une seule formule à programmer dans la cellule **B4**, puis on recopie à droite et en bas.

	A	B	C	D	E	F	G	H	I	J	K
1	TABLE DE MULTIPLICATION										
2											
3		1	2	3	4	5	6	7	8	9	10
4	1	1	2	3	4	5	6	7	8	9	10
5	2	2	4	6	8	10	12	14	16	18	20
6	3	3	6	9	12	15	18	21	24	27	30
7	4	4	8	12	16	20	24	28	32	36	40
8	5	5	10	15	20	25	30	35	40	45	50
9	6	6	12	18	24	30	36	42	48	54	60
10	7	7	14	21	28	35	42	49	56	63	70
11	8	8	16	24	32	40	48	56	64	72	80
12	9	9	18	27	36	45	54	63	72	81	90
13	10	10	20	30	40	50	60	70	80	90	100